Problem:
Score : 300 points

Problem Statement

Tak has N cards. On the i-th (1 \leq i \leq N) card is written an integer x_i.
He is selecting one or more cards from these N cards, so that the average of the integers written on the selected cards is exactly A.
In how many ways can he make his selection?

Constraints

1 \leq N \leq 50

1 \leq A \leq 50

1 \leq x_i \leq 50

N,\,A,\,x_i are integers.

Partial Score

200 points will be awarded for passing the test set satisfying 1 \leq N \leq 16.

Input

The input is given from Standard Input in the following format:

N A
x_1 x_2 ... x_N

Output

Print the number of ways to select cards such that the average of the written integers is exactly A.

Sample Input 1

4 8
7 9 8 9

Sample Output 1

5

The following are the 5 ways to select cards such that the average is 8:

Select the 3-rd card.

Select the 1-st and 2-nd cards.

Select the 1-st and 4-th cards.

Select the 1-st, 2-nd and 3-rd cards.

Select the 1-st, 3-rd and 4-th cards.

Sample Input 2

3 8
6 6 9

Sample Output 2

0

Sample Input 3

8 5
3 6 2 8 7 6 5 9

Sample Output 3

19

Sample Input 4

33 3
3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3 3

Sample Output 4

8589934591

The answer may not fit into a 32-bit integer.

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import           Data.Array.IArray (Array, listArray, range, (!))
import qualified Data.ByteString.Char8 as B
import           Data.Char
import           Data.List

main :: IO ()
main = do
    [n, a] <- readInts
    xs     <- listArray (1, n) <$> readInts
    print $ solve n a xs

readInts :: IO [Int]
readInts = unfoldr (B.readInt . B.dropWhile isSpace) <$> B.getLine

solve :: Int -> Int -> Array Int Int -> Int
solve n a xs = dps ! (n, n * xm) - 1
  where
    dp :: Int -> Int -> Int
    dp 0 t | t == n * xm = 1
           | otherwise   = 0
    dp j t | t - ys ! j < 0          = dps ! (j - 1, t)
           | t - ys ! j > 2 * n * xm = dps ! (j - 1, t)
           | otherwise               = dps ! (j - 1, t) + dps ! (j - 1, t - ys ! j)

    dps :: Array (Int, Int) Int
    dps      = listArray dpBounds [ dp j t | (j, t) <- range dpBounds ]

    dpBounds = ((0, 0), (n, 2 * n * xm))

    xm       = a `max` maximum [ xs ! i | i <- range (1, n) ]

    ys       = fmap (subtract a) xs